Министерство Российской Федерации по связи и информатизации

Сибирский Государственный Университет Телекоммуникаций и 

Информатики

Межрегиональный центр переподготовки специалистов

Курсовая работа

по дисциплине «Теория языков программирования и методы трансляции»

                                                  Выполнила: Русина О.Н.
                                                  Группа: ПБВ-52
                                                  Вариант: 17 

                                                  Проверила: Бах О.А.
Вы фамилию автора неверно указываете! В программе другая стоит.
Поскольку работа чужая, то для получения положительной оценки придётся выполнить доп. задание. 
Преобразовать заданную грамматику к БНФ. Вручную. И привести скриншот – как её преобразует программа. Если сначала с грамматикой нужно что-то сделать для того, чтобы можно было подавать её на вход программы, то добавьте эти дополнительные преобразования в программу. Грамматика порождает цепочки из скобочек:

Новосибирск, 2017
1. ПОСТАНОВКА ЗАДАЧИ
Написать программу для автоматического приведения заданной контекстно-свободной грамматики (КС-грамматики) к нормальной форме Хомского (БНФ).

Вход программы: терминальный и нетерминальный алфавиты грамматики, целевой символ, правила грамматики, 2 числа - диапазон длин для генерации цепочек.

Выход: построенная грамматика в БНФ (все 4 элемента), результат генерации цепочек по обеим грамматикам.
Подробно:

Язык задан КС-грамматикой, причём для приведения к БНФ она должна находиться в каноническом виде (раздел лекций 3.2.2). Приводить её к этому виду не требуется, достаточно только проверить корректность задания - действительно ли исходная грамматика находится в каноническом виде - и при отрицательном результате выдать соответствующее сообщение. Причём в этом сообщении должны быть конкретно указаны причины, почему именно грамматика не имеет канонического вида (например: <в грамматике присутствуют цепные правила: А>С>, или <в грамматике присутствует недостижимый символ: В>). Для того чтобы в исходной грамматике можно было использовать пустое правило, необходимо либо предусмотреть поле ввода для символа, которым обозначается пустая цепочка, либо дать пояснения пользователю, как именно ему следует задавать пустое правило.

Программа должна:

1.проверить заданную КС-грамматику - находится ли она в каноническом виде и при отрицательном результате выдать сообщение;
2.привести заданную КС-грамматику к нормальной форме Хомского (раздел 3.3.1);
3.проверить построенную грамматику (БНФ) на эквивалентность исходной.
Для проверки построенной грамматики в БНФ на эквивалентность исходной по обеим грамматикам следует сгенерировать множества всех цепочек в заданном пользователем диапазоне длин и проверить эти множества на идентичность. Генерация цепочек осуществляется в соответствии с лабораторной работой №1.
При обнаружении несовпадения должна выдаваться диагностика различий - где именно несовпадения и в чём они состоят. Для удобства сравнения множества цепочек необходимо упорядочить, цепочки перенумеровать, для генерации цепочек по каждой грамматике сделать отдельную кнопку. Следует предусмотреть возможность многократного изменения диапазона длин цепочек и повторной их генерации для новых значений длины.

Приведение КС-грамматики к нормальной форме Хомского следует осуществлять согласно алгоритму, изложенному в лекционном материале (раздел 3.3.1).
2. АЛГОРИТМ РЕШЕНИЯ ЗАДАЧИ

Входными данными для программы являются КС-грамматика и минимальная и максимальная длина генерируемых цепочек. КС-грамматика может быть введена с клавиатуры или из заранее подготовленного файла. При вводе в клавиатуры необходимо заполнить соответствующие поля на экранной форме. Все поля сопровождаются описаниями информации,  которую они должны содержать и ввод данных в них не требует специальных пояснений. Полей данных четыре:

   -множество нетерминальных символов;

   -множество терминальных символов;

   -целевой символ;

   -правила грамматики;

При подготовке КС-грамматики в файле в его первой строке перечисляются нетерминальные символы без разделителей между ними. Во второй строке перечисляются терминальные символы без разделителей между ними. Третья строка должна содержать нетерминальный символ, являющийся целевым. В строках с четвертой и далее записываются правила грамматики в виде:

Нетерминальный символ=строка терминальных и нетерминальных символов

          Для задания пустой цепочки используется специальный символ @.

Описание структуры файла доступно в программе в диалоговом режиме путем использования меню «Помощь».

При выводе цепочек по КС-грамматике или по БНФ программа запрашивает ввод с клавиатуры:

-минимальную длину цепочек, которая должна быть больше или равна 0;

-максимальную длину цепочек.

Выходными данными программы являются построенная БНФ и два списка цепочек, первый из которых содержит цепочки, сгенерированные по КС-грамматике, второй – по БНФ. Оба списка цепочки одинаково упорядочены и снабжены порядковыми номерами.  Выходные данные отображаются в трех отдельных окнах, что позволяет пользователю программы настроить их размер и расположение на экране удобным для него образом. 

Перед преобразованием КС-грамматики в БНФ программа проверяет корректность введенных данных:

-наличие целевого символа во множестве нетерминальных символов грамматики;

-наличие нетерминалов в левой части правил во множестве нетерминальных символов грамматики;

-наличие во всех правилах грамматики символа =, разделяющего левую и правую части правил;

-отсутствие в правых частях правил символов, не входящих во множество терминальных и нетерминальных символов.
Далее программа проверяет соответствие правил КС-грамматики каноническому виду. Для этого используются алгоритмы, изложенные в разделе лекций 3.2.2.  Они предназначены для удаления из грамматики пустых и цепных правил, а также бесплодных и недостижимых символов. Поскольку по условию варианта приведение КС-грамматики к каноническому виду не требуется, то эти алгоритмы реализованы не полностью. Они лишь проверяют факт наличия правила или символа нарушающего канонический вид грамматики.
КС-грамматика отвечающая всем требованиям канонического вида преобразуется к БНФ по алгоритму, изложенному в разделе лекций 3.3.1.

Пример преобразования КС-грамматики языка арифметических выражения из скобок, переменных a и b, операций + и -:
G({a,b,(,),+,-},{E},P,E)

правила P:
E(a

E(b
E((E)

E(E+E
E(E-E
1) Переносим во множество нетерминальных символов БНФ все нетерминальные символа КС-грамматики: VN(={E}
2) В соответствии с пунктом 1 алгоритма правило E(a переносится в  БНФ без изменений.
3) В соответствии с пунктом 1 алгоритма правило E(b переносится в  БНФ без изменений.
4) Правило E((E) соответствует пункту 5 алгоритма, когда правая часть правила является цепочкой терминальных и нетерминальных символов длиной более 2. Добавляем новый нетерминальный символ в VN(={E,A}, в правила БНФ добавляем A((. Добавляем два новых нетерминальных символа в VN(={E,A,B,C}, в правила БНФ добавляем B(EC и C(). Исходное правило преобразуем к виду: E(AB.
5) Правило E(E+E соответствует пункту 5 алгоритма. Добавляем два новых нетерминальных символа в VN(={E,A,B,C,D,F}, в правила БНФ добавляем D(FE и F(+. Исходное правило преобразуем к виду: E(ED.
6) Правило E(E-E соответствует пункту 5 алгоритма. Добавляем два новых нетерминальных символа в VN(={E,A,B,C,D,F,G,H}, в правила БНФ добавляем G(HE и H(-. Исходное правило преобразуем к виду: E(EG.
Преобразование КС-грамматики в БНФ выполнено:
G( ({a,b,(,),+,-},{E,A,B,C,D,F,G,H},P(,E)

правила P(:

E(a

E(b

Для генерации цепочек КС-грамматики и БНФ используется процедура левостороннего вывода, реализованная с использованием программных блоков лабораторной работы №1. 

3. ОПИСАНИЕ ОСНОВНЫХ ПЕРЕМЕННЫХ, БЛОКОВ И ПОДПРОГРАММ

Основные переменные программы:

-TGrammatic – структура, использующаяся для хранения КС-грамматики (KS) и БНФ (BNF).
Поля структуры:
-VN – множество нетерминальных символов;

-Rules – массив правил грамматики;
-LRules – массив минимальных длин цепочек, выводимых из каждого нетерминальных символов грамматики;
-NRules – массив количества правил грамматики для каждого нетерминального символа.

Простые переменные:

-VT – множество терминальных символов КС-грамматики и БНФ;

-S0 – начальный символ КС-грамматики и БНФ;

-LMax – максимальная длина генерируемых цепочек;

-LMin – минимальная длина генерируемых цепочек.

Основные программные блоки:

-LRules – вычисление минимальных длин цепочек, выводимых из нетерминальных символов;
-Canonic – проверка наличия в КС-грамматики бесплодных и недостижимых символов и цепных правил;

-InputKS – ввод КС-грамматики с экранной формы с проверкой корректности данных и отсутствия пустых правил;

-GenerateStr – процедура генерации цепочек;
-KS_BNF – процедура приведения КС-грамматики к БНФ;
4. ТЕКСТ ПРОГРАММЫ
Unit1.pas
unit Unit1;

interface

uses

  Windows, Messages, SysUtils, Variants, Classes, Graphics, Controls, Forms,

  Dialogs, Math, StdCtrls, Grids, Menus, ExtCtrls;

type

  TGrammatic=Record

   Rules: Array [1..26,1..20] of String;

   LRules: Array [1..26] of Integer;

   NRules: Array [1..26] of Integer;

   VN: String;

  end;

  TForm1 = class(TForm)

    Label1: TLabel;

    MainMenu1: TMainMenu;

    N1: TMenuItem;

    N3: TMenuItem;

    N4: TMenuItem;

    N5: TMenuItem;

    N6: TMenuItem;

    Memo1: TMemo;

    N2: TMenuItem;

    N7: TMenuItem;

    SaveDialog1: TSaveDialog;

    N8: TMenuItem;

    N9: TMenuItem;

    N10: TMenuItem;

    OpenDialog1: TOpenDialog;

    N12: TMenuItem;

    ListBox3: TListBox;

    Label2: TLabel;

    Label3: TLabel;

    Edit1: TEdit;

    Label4: TLabel;

    Edit2: TEdit;

    N13: TMenuItem;

    Label5: TLabel;

    Edit3: TEdit;

    Label6: TLabel;

    procedure N6Click(Sender: TObject);

    procedure N3Click(Sender: TObject);

    procedure N5Click(Sender: TObject);

    procedure N4Click(Sender: TObject);

    procedure N2Click(Sender: TObject);

    procedure N7Click(Sender: TObject);

    procedure N9Click(Sender: TObject);

    procedure N10Click(Sender: TObject);

    procedure N12Click(Sender: TObject);

  private

    function NonTerminal: String;

    function Rule(s: String; ivn: integer): String;

    function Canonic: Boolean;

    procedure LRules(var g: TGrammatic);

    procedure GenerateStr(var g: TGrammatic; s: String; depth: integer; ListBox: TListBox);

    function InputKS: Boolean;

    function KS_BNF: Boolean;

  public

    { Public declarations }

  end;

var

 Form1: TForm1;

 BNF, KS: TGrammatic;

 VT, S0: String;

 LMax, LMin: Integer;

implementation

uses Unit2, Unit3, Unit4;

{$R *.dfm}

{вычисление длин минимальных цепочек, выводимых из каждого нетерминала грамматики}

procedure TForm1.LRules(var g: TGrammatic);

var I, J, K, L, N: Integer;

begin

 for I:=1 to Length(g.VN) do g.LRules[I]:=-1;

 for I:=1 to Length(g.VN) do with g do begin

  if LRules[I]<>-1 then continue;

  for J:=1 to NRules[I] do begin

   L:=0;

   if Rules[I][J]='@' then begin

    LRules[I]:=0;

    break;

   end

   else

    for K:=1 to Length(Rules[I,J]) do begin

     N:=Pos(Rules[I,J][K],VN);

     if N=0 then Inc(L)

     else

      if LRules[N]<>-1 then Inc(L,LRules[N])

      else begin

       L:=-1;

       break;

      end;

    end;

   if (L>=0) And ((LRules[I]<0) Or (LRules[I]>L)) then LRules[I]:=L;

  end;

 end;

 for I:=1 to Length(g.VN) do with g do

  if LRules[I]=-1 then LRules[I]:=1;

end;

{формирование строки описания грамматики}

function GrammaticStr(VT, VN, S0: String): String;

var S: String; I: Integer;

begin

 S:='G({';

 for I:=1 to Length(VT) do begin

  S:=S+VT[I];

  if I<>Length(VT) then S:=S+','

                   else S:=S+'},{';

 end;

 for I:=1 to Length(VN) do begin

  S:=S+VN[I];

  if I<>Length(VN) then S:=S+','

                   else S:=S+'},P,';

 end;

 S:=S+S0+')';

 GrammaticStr:=S;

end;

{добавление нового нетерминального символа в грамматику}

function TForm1.NonTerminal: String;

var s: String;

begin

 with BNF do begin

  s:='A';

  while (Pos(s,VN)<>0) do Inc(s[1]);

  VN:=VN+s;

 end;

 Result:=s;

end;

{добавление правила s для нетерминала ivn}

function TForm1.Rule(s: String; ivn: integer): String;

var v: String; i, n: integer;

begin

 with BNF do begin

  n:=Length(VN);

  if Length(s)=1 then

   for i:=ivn+1 to n do

    if Rules[i][1]=s then begin

     Result:=VN[i];

     exit;

    end;

  v:=NonTerminal;

  Inc(n);

  NRules[n]:=1;

  Rules[n][1]:=s;

 end;

 Result:=v;

end;

{проверка КС-грамматики на цепные правила и бесплодные и недостижимые символы}

function TForm1.Canonic: Boolean;

var s, s1, s2: String; i, j, k: integer;

begin

 with KS do

  for i:=1 to Length(VN) do

   for j:=1 to NRules[i] do

    if (Length(Rules[i][j])=1) And (Pos(Rules[i][j],VN)>0) then begin

     ShowMessage('Цепное правило в КС-грамматике: '+VN[i]+'='+Rules[i][j]);

     Result:=false;

     exit;

    end;

 s1:='';

 s2:='';

 while TRUE do with KS do begin

  for i:=1 to Length(VN) do

   for j:=1 to NRules[i] do begin

    if Pos(VN[i],s2)>0 then continue;

    s:=Rules[i][j];

    for k:=1 to Length(s) do

     if (Pos(s[k],VN)>0) And (Pos(s[k],s1)=0) then begin

      s2:=s2+VN[i];

      break;

     end;

   end;

  if s1=s2 then break;

  s1:=s2;

 end;

 for i:=1 to Length(KS.VN) do

  if Pos(KS.VN[i],s1)=0 then begin

   ShowMessage('В КС-грамматике бесплодный символ '+KS.VN[i]);

   Result:=false;

   exit;

  end;

 s1:=S0;

 s2:=S0;

 while true do with KS do begin

  for i:=1 to Length(VN) do begin

   if Pos(VN[i],s1)=0 then continue;

   for j:=1 to NRules[i] do begin

    s:=Rules[i][j];

    for k:=1 to Length(s) do

     if (Pos(s[k],VN)<>0) And (Pos(s[k],s2)=0) then begin

      s2:=s2+s[k];

      break;

     end;

   end;

  end;

  if s1=s2 then break;

  s1:=s2;

 end;

 for i:=1 to Length(KS.VN) do

  if Pos(KS.VN[i],s1)=0 then begin

   ShowMessage('В КС-грамматике недостижимый символ '+KS.VN[i]);

   Result:=false;

   exit;

  end;

 Result:=true;

end;

{рекурсивный левосторонний вывод цепочек}

procedure TForm1.GenerateStr(var g: TGrammatic; s: String; depth: integer; ListBox: TListBox);

var sl, sr: String; i, j, k, l, m, n: integer;

begin

 if depth>50 then exit;

 with g do begin

  i:=1;

  l:=Length(s);

  while (i<=l) do begin

   n:=Pos(s[i],VN);

   if n<>0 then break

           else Inc(i);

  end;

  if i>l then begin

   if (l>=lmin) And (l<=lmax) then begin

    if s='' then s:='@';

    ListBox.Items.Add(s);

   end;

   exit;

  end;

  k:=i-1;

  for j:=i to Length(s) do begin

   m:=Pos(s[j],VN);

   if m=0 then Inc(k)

          else Inc(k,LRules[m]);

  end;

  if k>lmax then exit;

  Inc(depth);

  sl:=Copy(s,1,i-1);

  sr:=Copy(s,i+1,l-i);

  for i:=1 to NRules[n] do

   if Rules[n][i]='@' then GenerateStr(g,sl+sr,depth,ListBox)

                      else GenerateStr(g,sl+Rules[n][i]+sr,depth,ListBox);

 end;

end;

{ввод КС-грамматики с формы с проверкой на отсутствие пустых правил}

function TForm1.InputKS: Boolean;

var s: String; i, j, n: integer;

begin

 with KS do begin

  VN:=Edit1.Text;

  VT:=Edit2.Text;

  S0:=Edit3.Text;

  if Pos(S0,VN)=0 then begin

   ShowMessage('Целевого символа нет во множестве нетерминальных символов');

   Result:=false;

   exit;

  end;

  for i:=1 to 26 do NRules[i]:=0;

  for i:=0 to Memo1.Lines.Count-1 do begin

   s:=Memo1.Lines.Strings[i];

   if s='' then continue;

   n:=Pos(s[1],VN);

   if n=0 then begin

    ShowMessage('Нетерминал '+s[1]+' не определен');

    Result:=false;

    exit;

   end;

   j:=Pos('=',s);

   if j=0 then begin

    ShowMessage('Нет = после левой части правила');

    Result:=false;

    exit;

   end;

   s:=Copy(s,j+1,Length(s));

   for j:=1 to Length(s) do

    if s[j]='@' then begin

     if n<>Pos(S0,VN) then begin

      ShowMessage('Пустое правило допустимо только для начального символа');

      Result:=false;

      exit;

     end;

    end

    else

     if (Pos(s[j],VN)=0) And (Pos(s[j],VT)=0) then begin

      ShowMessage('В правиле посторонний символ '+s[j]);

      Result:=false;

      exit;

     end;

   Inc(NRules[n]);

   Rules[n][NRules[n]]:=s;

  end;

 end;

 Result:=true;

end;

{преобразование КС в БНФ}

function TForm1.KS_BNF: Boolean;

var rnew, s, v, vnew: String; i, j, l, n: integer;

begin

 Form2.ListBox1.Items.Clear;

 for i:=1 to 26 do bnf.NRules[i]:=0;

 bnf.VN:=ks.VN;

 l:=Length(ks.VN);

 for n:=1 to l do

  for i:=1 to ks.NRules[n] do begin

   v:=ks.VN[n];

   s:=ks.Rules[n][i];

   while (Length(s)>2) do begin

    if Pos(s[1],VT)<>0 then rnew:=Rule(s[1],l)

                       else rnew:=s[1];

    vnew:=NonTerminal;

    rnew:=rnew+vnew;

    j:=Pos(v,bnf.VN);

    Inc(bnf.NRules[j]);

    bnf.Rules[j][bnf.NRules[j]]:=rnew;

    v:=vnew;

    Delete(s,1,1);

   end;

   if Length(s)=1 then begin

    j:=Pos(v,bnf.VN);

    Inc(bnf.NRules[j]);

    bnf.Rules[j][bnf.NRules[j]]:=s;

   end

   else begin

    if Pos(s[1],VT)<>0 then begin

     rnew:=Rule(s[1],l);

     vnew:=s[2];

     s:=rnew+vnew;

    end;

    if Pos(s[2],VT)<>0 then begin

     rnew:=Rule(s[2],l);

     vnew:=s[1];

     s:=vnew+rnew;

    end;

    j:=Pos(v,bnf.VN);

    Inc(bnf.NRules[j]);

    bnf.Rules[j][bnf.NRules[j]]:=s;

   end;

  end;

 for i:=1 to Length(bnf.VN) do with bnf do

  for j:=1 to NRules[i] do

   Form2.ListBox1.Items.Add(VN[i]+'='+Rules[i][j]);

 Form2.Label1.Caption:=GrammaticStr(VT,BNF.VN,S0);

 Form2.Show;

 Result:=true;

end;

procedure TForm1.N6Click(Sender: TObject);

begin

 ShowMessage('Автор: Гилязов Алексей Нургаясович, гр.ПБТ-22');

end;

procedure TForm1.N3Click(Sender: TObject);

begin

 ShowMessage('Приведение контекстно-свободной грамматики в канонической форме к нормальной форме Хомского');

end;

procedure TForm1.N5Click(Sender: TObject);

begin

 ShowMessage(

  '           Описание входного файла с грамматикой'+Chr(13)+Chr(13)+

  'В первой строке перечисляются нетерминальные символы без разделителей между ними.'+

  'Во второй строке перечисляются терминальные символы без разделителей между ними.'+

  'Третья строка должна содержать нетерминальный символ, являющийся целевым.'+

  'В строках с четвертой и далее записываются правила грамматики в виде:'+Chr(13)+Chr(13)+

  'Нетерминальный символ=строка терминальных и нетерминальных символов'+Chr(13)+Chr(13)+

  'Пустая цепочка задается символом @'

 );

end;

procedure TForm1.N4Click(Sender: TObject);

begin

 if Not InputKS then exit;

 if Canonic And KS_BNF then N7.Enabled:=true;

end;

procedure TForm1.N2Click(Sender: TObject);

var s: String; i: integer;

begin

 if Not InputKS then exit;

 s:=InputBox('Ввод данных для генерации цепочек','Введите минимальную длину','');

 if (Not TryStrToInt(s,lmin)) Or (lmin<0) then begin

  ShowMessage('Минимальная длина цепочек должна не меньше 0');

  exit;

 end;

 s:=InputBox('Ввод данных для генерации цепочек','Введите максимальную длину','');

 if (Not TryStrToInt(s,lmax)) Or (lmax<lmin) then begin

  ShowMessage('Ошибка в задании максимальной длины цепочек');

  exit;

 end;

 Form3.ListBox1.Items.Clear;

 Form3.ListBox1.Sorted:=True;

 Label6.Visible:=true;

 Form1.Update;

 LRules(KS);

 GenerateStr(KS,S0,0,Form3.ListBox1);

 I:=0;

 while (I<Form3.ListBox1.Items.Count-1) do

  if (Form3.ListBox1.Items.Strings[I]=Form3.ListBox1.Items.Strings[I+1]) then

   Form3.ListBox1.Items.Delete(I+1)

  else

   Inc(I);

 for I:=0 to Form3.ListBox1.Items.Count-1 do

  Form3.ListBox1.Items.Strings[I]:=IntToStr(I+1)+') '+Form3.ListBox1.Items.Strings[I];

 N9.Enabled:=true;

 Label6.Visible:=false;

 Form3.Show;

end;

procedure TForm1.N7Click(Sender: TObject);

var S: String; I: integer;

begin

 s:=InputBox('Ввод данных для генерации цепочек','Введите минимальную длину','');

 if (Not TryStrToInt(s,lmin)) Or (lmin<0) then begin

  ShowMessage('Минимальная длина цепочек должна не меньше 0');

  exit;

 end;

 s:=InputBox('Ввод данных для генерации цепочек','Введите максимальную длину','');

 if (Not TryStrToInt(s,lmax)) Or (lmax<lmin) then begin

  ShowMessage('Ошибка в задании максимальной длины цепочек');

  exit;

 end;

 Form4.ListBox1.Items.Clear;

 Form4.ListBox1.Sorted:=True;

 Label6.Visible:=true;

 Form1.Update;

 LRules(BNF);

 GenerateStr(BNF,S0,0,Form4.ListBox1);

 I:=0;

 while (I<Form4.ListBox1.Items.Count-1) do

  if Form4.ListBox1.Items.Strings[I]=Form4.ListBox1.Items.Strings[I+1] then

   Form4.ListBox1.Items.Delete(I+1)

  else

   Inc(I);

 for I:=0 to Form4.ListBox1.Items.Count-1 do

  Form4.ListBox1.Items.Strings[I]:=IntToStr(I+1)+') '+Form4.ListBox1.Items.Strings[I];

 N10.Enabled:=true;

 Label6.Visible:=false;

 Form4.Show;

end;

procedure TForm1.N9Click(Sender: TObject);

begin

 SaveDialog1.InitialDir:=GetCurrentDir;

 if Not SaveDialog1.Execute then exit;

 Form3.ListBox1.Items.SaveToFile(SaveDialog1.FileName);

end;

procedure TForm1.N10Click(Sender: TObject);

begin

 SaveDialog1.InitialDir:=GetCurrentDir;

 if Not SaveDialog1.Execute then exit;

 Form4.ListBox1.Items.SaveToFile(SaveDialog1.FileName);

end;

procedure TForm1.N12Click(Sender: TObject);

var I: Integer;

begin

 OpenDialog1.InitialDir:=GetCurrentDir;

 if Not OpenDialog1.Execute then exit;

 ListBox3.Items.LoadFromFile(OpenDialog1.FileName);

 Memo1.Lines.Clear;

 with KS do begin

  Edit1.Text:=ListBox3.Items.Strings[0];

  Edit2.Text:=ListBox3.Items.Strings[1];

  Edit3.Text:=ListBox3.Items.Strings[2];

  for I:=3 to ListBox3.Items.Count-1 do

   Memo1.Lines.Add(ListBox3.Items.Strings[I]);

 end;

 N7.Enabled:=FALSE;

end;

end.
5. РЕЗУЛЬТАТЫ ТЕСТИРОВАНИЯ ПРОГРАММЫ

Первоначальная отладка и тестирование программы были выполнены на  КС-грамматике, описанной в примере в разделе 2 отчета. В целях отсутствия ошибок при многократных вводах данных в программу, описание грамматики было записано в файл ks-grammatic.txt. Результат работы отлаженной программы приведен на экране ниже. Для генерации цепочек был указан диапазон длин от 1 до 7.

Далее тестирование программы было выполнено на КС-грамматике, преобразование которой к БНФ приведено в качестве примера в разделе лекций 3.3.1. БНФ, построенная программой, совпадает с БНФ, приведенной в лекциях с точностью до обозначения нетерминальных символов. Соответствие нетерминальных символов из примера символам из программы следующее:

<aB>

D
<a>

E
<b>

F
<c>

G
На экране приведены окончания списков цепочек, сгенерированных в диапазонах длин от 3 до 10. Количество и состав цепочек в обоих случаях совпадают.

